

FOXPRO2.5 系统程序设计(三)

罗 辉 (湖南省双峰工行) 何丹鸿 (湖南省娄底人行)

第四讲 数据库浏览命令 Browse 的使用

一、数据库浏览命令 Browse

Browse 命令在 FoxPro 系统中是一个最有用的命令。它一改在 FOXBASE 中呆板的使用方式,提供大量的功能子句,使你可以灵活方便地构筑出丰富多样功能强大的编辑、查询画面,起到事半功倍之效。

1. Browse 窗口的使用

使用 Browse 命令将打开一个 Browse 窗口,并将数据库中的记录在该窗口上显示出来;同时在系统菜单棒上附加一个 Browse 菜单基。窗口顶部是 Browse 头,默认时,它由所编辑的各字段名组成,而且以它们在数据库结构中出现的顺序和宽度在 Browse 窗口是从左到右排列显示的。你可用 Browse 菜单中的 Size field 和 Move field 选项改变其显示宽度和排列顺序;也可用你所定义的其它字符串取代该字段名的显示位置,它只用于 Browse 头的显示,并不影响该字段在数据库结构中的名。在这一窗口上你可以编辑和追加记录。只要有足够的内存,也可以为每个工作区打开一个 Browse 窗口。

在 Browse 窗口上,用上移(↑)、下移(↓)或窗口的垂直滚动条一条条记录的移动,或用翻页键 PgUp 或 PgDn 一屏屏的翻阅;而用 Tab 或 Shift+Tab 键则可实现字段编辑域的左右移动。

按 Ctrl+W 或 Ctrl+End 键或点按关闭框可以存储在 Browse 窗口中所作的改变,并退出 Browse 窗口;如果按 Ctrl+Q 或 Escape,也将退出 Browse 窗口,但当前字段编辑域所作的编辑将不作保存;在程序中,也可使用 DEACTIVATE WINDOW <Browse 窗口名> 命令存储所作的改变并关闭 Browse 窗口。

2. 备注字段和通用字段的编辑

如果要在 Browse 窗口编辑备注型字段域,只需移光标

到该域,然后按 Ctrl+PgDn 键或双点按备注字段即可进入备注字段编辑窗口。在备注字段编辑窗口中,按 Ctrl+W 或点按关闭框,将退出并保存所作的修改,按 Esc 键将不保存改变而退出。

同样在 Browse 窗口,定位到以“OLE 连接”的通用字段上,按 Ctrl+PgDn 键或双点按该字段,即可打开该字段内容所对应的源实用程序编辑窗口对其进行编辑。

3. Browse 处理多库

用 FoxPro 可以在多个库间建立关联。这时当记录指针在“父”库中移动时,“子”库中指针将自动指向相应的记录。你可以同时打开一个父 Browse 窗口和多个相关的子库。

在同一个 Browse 窗口中还可同时编辑父库和多个关联子库的多个字段域。先选择父库所在工作区,然后执行带 FIELDS 的 Browse 命令。在 FIELDS 后的字段列表中,子库的字段名必须带上别名。

4. Browse 处理一对多关联

在相关联的两个库间,父库中的每个记录,在子库中可以有多个相关联的记录。用 SET SKIP 命令可以在两个库间建立一对多关联。

对显示在 Browse 窗口的父库的某一记录,子库中相关的多个记录都可以显示出来。父记录与子库中第一个匹配的记录一起显示。随后其它匹配的记录则显示在父记录和第一个匹配的子记录之后的列上。

如果记录指针正指在父记录上,可通过按 Ctrl+↓(↑)键移到下(上)一个父记录。

【例 3.1】用 SET SKIP 命令在两个库间建立一个一对多关联。对库 OFFICES 中每个记录,在库 CUSTOMER 中可以对对应多个记录。关联建立之后,将打开一个同时显示父 / 子库中记录的 Browse 窗口。

CLEAR

CLOSE DATABASES

USE offices INDEX offices ORDER ono IN 0 &&父库

USE customer INDEX customer ORDER ono IN 0&&子

库

SELECT offices 回到父工作区

SET RELATION TO ono INTO customer &&建立关联

SET SKIP TO customer &&一对多关联

WAIT WINDOW 'Scroll to see companies for each office'

NOWAIT

Browse FIELDS offices.ono :H='办事处号',;

offices.city :H='办事处所在城市', customer.company

5. 与 Browse 应用有关的函数

有几个能返回关于 Browse 窗口信息的 FoxPro 函数, 用于在 Browse 命令有关子句的设计中。

VARREAD()函数返回在 Browse 窗口中当前正被编辑的字段名。

RECNO()函数当前记录的记录号。

ROW()和 COL()则可返回当前光标在 Browse 中的行列位置。

二、Browse 命令的使用

在 Browse 窗口提供的 Browse 菜单中有许多实用灵活的功能可方便地对 Browse 窗口进行控制和操作。但是命令(程序)级的 Browse 则提供许多功能子句, 能实现比 Browse 菜单更为灵活丰富的功能。

1. Browse 命令的使用格式

Browse [FIELDS <field list>]

[FONT <expC1> [, <expN1>]] [STYLE <expC2>]

[FOR <expL1>] [FORMAT] [FREEZE <field>]

[KEY <expr1> [, <expr2>]] [LAST]

[LEDIT] [REDIT] [LOCK <expN2>] [LPARTITION]

[NOAPPEND] [NOCLEAR] [NODELETE] [NOEDIT]

NOMODIFY]

[NOLGRID] [NORGRID] [NOLINK] [NOMENU]

[NOOPTIMIZE] [NOREFRESH] [NORMAL]

[NOWAIT]

[PARTITION <expN3>] [PREFERENCE <expC3>]

[REST] [SAVE] [TIMEOUT <expN4>] [TITLE

<expC4>]

[VALID [:F] <expL2> [ERROR <expC5>]] [WHEN <expL3>]

[WIDTH <expN5>] [[WINDOW <window name1>]

[IN [WINDOW] <window name2> IN SCREEN]]

[COLOR SCHEME <expN6> COLOR <color pair

list>]

2. Browse 命令的可选子句及其使用

(1) FIELDS <field list> FIELDS 子句将按字段列表

中所列的字段名及其顺序在 Browse 窗口显示, 未列入的字段将不被显示和编辑。如不带该子句, 将按数据库结构中字段的物理顺序显示全部字段。它可包含多个相关库的字段, 但须带上别名。

字段列表 <field list> 的使用格式是:

<字段一> [:R] [:列宽度]

[:V = <exp1> [:F] [:E = <expC1>]]

[:P = <expC2>] [:B = <expr2>, <expr3> [:F]]

[:H = <expC3>] [:W = <expL1>]

[, <字段二> [:R]...]

九个字段级可选项的意义如下:

①:R表明该字段只能读不能编辑。

②: <列宽度>表明该字段在 Browse 窗口中的显示宽度。它并不影响数据库中的实际字段宽度。

③:V = <exp1> 字段级校验项。当将光标从该字段编辑域移走时, 如果 <exp1> 值为真, 系统认为所编辑的数据是合法的, 允许移走光标; 如果 <exp1> 值为假, 系统认为所输入的数据不合法, 光标将仍停留在该字段上, 并显示一条错误信息; 如果 <exp1> 值为 0, 系统认为输入数据不合法, 光标仍停在该字段, 但不显示错误信息。默认情况下, 只有当字段值被修改了才计算表达式 <exp1>, 你也可带选项[:F]强制校验。用选项[:E]则可描述不同的错误信息。校验项对备注字段和通用字段无效。

④:F不论该字段修改与否, 强制校验。⑤:E = <expC1> 用选项[:V]校验输入数据非法时, 默认系统错误信息为 "Invalidinput"。用[:E]可指定错误显示信息为 <expC1>。但只有当 SET NOTIFY 为 ON 时, 该信息才显示; 如果此时 SET BELL 为 ON, 系统还会鸣响。

【例 3.2】打开数据库 CUSTOMER, 显示 CNO 和 COMPANY 两个字段。字段 CNO(顾客号)是一 5 字节宽的字符字段, CNO 必须大于 "14000"。COMPANY 编辑域显示宽度设为 10 个字符。

CLOSE DATABASES

IF WINDOWS

SET STATUS BAR ON

ENDIF

USE customer

Browse FIELDS cno:V = cno > '14000':F;

:E='顾客号须大于 14000', company:10

⑥:P = <expC2> 图像选项,用于Browse窗口中字段数据的输入和显示。<expC2>是控制代码串,其有效控制代码及其使用请参阅有关资料。

⑦:B = <expr2>, <expr3> [F] 边界校验。如果输入数据不落在表达式<expr2>和<expr3>的值之间,则数据非法,系统显示一错误信息指示数据的合法范围。表达式<expr2>和<expr3>不能是用户自定义函数。默认情况下,只有当字段值被修改了才进行范围校验,也可带选项[F]强制校验。

【例 3.3】销售记录输入,其值在 100~99,999 之间,使用控制代码控制只能输入数字。CLOSE DATABASES

```
USE salesman
```

```
Browse FIELDS ytdsales :P = '99999';
```

```
:B = 100, 99999 :F, name
```

⑧:H = <expC3> 字段显示名定义。默认时, Browse 窗口的数据录入标题【Browse 头沿用库中各字段名。通过本项可定义该字段的显示名。FOXPRO2.5 不能处理汉字字段名,用该选项可使英文字段名以汉字形式显示在 Browse 上。

【例 3.4】对英文数据库字段,在 Browse 窗口以对应的汉字字段名显示出来。注意相应的汉字名字符串前最好有一个空格,不然在有些中西文汉字系统下,该串第一个汉字的内码将与 Browse 窗口固有的分隔各字段的制表符在一起,产生二义性,显示为乱码。

```
Browse FIELDS ytdpurch :H = '购买品';
```

```
company :H = '购买单位'
```

⑨:W = <expL1> 字段级编辑权验证。根据逻辑表达式<expL1>的值为真(假),允许(禁止)光标移到该字段编辑域。表达式<expL1>支持用户自定义函数。

注意:

字段列表中不一定必须是真实的数据库字段名。

Browse 允许使用“计算字段”的虚字段名,它容纳将有关字段计算后的结果信息。

计算字段包含由一个表达式建立的只读数据。这个表达式可以是任何格式,只要它是合法的 FoxPro 表达式。

建立计算字段的格式是: <计算字段名> = <expr>。

【例 3.5】产生一个叫 LOCATION 的计算字段。

```
CLOSE DATABASES
```

```
USE customer
```

```
Browse FIELD LOCATION = ALLTRIM(city) + ',' + state
```

其中 CITY 和 STATE 字段是被选择库中的真实字段。

(2)FONT <expC1> [, <expN1>] 和 STYLE <expC2>。FONT 定义字段编辑域的字体,<expC1>是字体名,<expN1>是字体大小;STYLE 定义编辑域字体风格。FONT 和 STYLE 只适应于 Foxpro 2.5 for Windows。

【例 3.6】字段编辑域使用 15 点阵带下划线的 SYSTEM 字体:

```
CLOSE DATABASES
```

```
USE customer
```

```
Browse FIELDS contact FONT 'System', 15 STYLE 'NU'
```

(3)FOR <expL1>。FOR 子句使 Browse 窗口有条件地显示和编辑满足逻辑表达式<expL1>的记录。最好使用优化的表达式,以启用 Rushmore 技术提高查询效率。

(4)FORMAT。本子句使 Browse 窗口的显示和数据录入受标明的格式文件的格式控制。该格式文件必须先用 SET FORMAT 命令打开。

格式文件中如下内容将对 Browse 窗口有效:所有字段列表、所有 VALID、WHEN 和 RANGE 子句、用 PICTURE 表明的字段编辑域大小、所有 SAY 子句等。

【例 3.7】用格式文件控制 Browse 窗口。

```
USE customer
```

```
SET FORMAT TO custentr.fmt
```

```
Browse FORMAT
```

相应格式文件 CUSTENTR.FMT 内容如下:

```
@ 3,0 GET cno PICTURE 'NNNNN'
```

```
@ 3,0 GET company VALID company != SPACE(40);  
PICTURE 'AAAAAAAAAAAAAAAAAAAAA'
```

```
@ 3,0 GET contact WHEN contact = SPACE(40)
```

```
@ 3,0 SAY ytdpurch
```

```
@ 3,0 SAY ALLTRIM(city) + ',' + ALLTRIM(state);  
PICTURE 'XXXXXXXXXXXXXXXXXXXXX'
```

(5)FREEZE <field>。该子句使 Browse 窗口只能编辑一个字段编辑域,其它域则只能显示不能编辑。

(6)KEY <expr1> [, <expr2>]。限制可处理的记录范围。用 KEY 子句表明索引键值或索引键值范围。此时数据库必须被索引,且索引关键词必须与 KEY 表明的键值

类型一致。

【例 3.8】只对邮政编码落在 43000-43999 范围内的销售记录的 YTDPURCH 字段进行编辑。CLOSE DATABASES

USE customer

SET ORDER TO zip &&此前已用邮政编码索引

Browse FIELDS ytdpurch :H = '购买品',;

company :H = '购买单位' FREEZE ytdpurch ;

KEY '43000', '43999'

(7)LAST. Browse 窗口被关闭时, 如果 SET RESOURCE 为 ON, 则任何对 Browse 窗口配置属性的改变都将保存到资源文件 FOXUSER 中, 如编辑的字段列表、大小、窗口的大小和位置等等。此后如果你带 LAST 子句执行 Browse 命令, 将使用上次保存的配置属性。

但如果上次 Browse 窗口是带 PREFERENCE 子句打开的, 则 Browse LAST 命令不能恢复其原来的配置属性。

按 Ctrl+Q 退出 Browse 窗口将不保持当前的配置修改。

(8)LEDIT 和 REDIT. Browse 窗口的信息能以两种方式显示: Change 或 Browse 方式。默认为 Browse 方式, 所有编辑字段水平显示。而 Change 方式字段将垂直显示。

可以用下列方式将 Browse 窗口分裂成左右两部分: 用鼠标拖曳分裂框、从 Browse 菜单基上选择 Resize Partitions 菜单项、或使用带 PARTITION 子句的 Browse 命令。左右两部分都可各自独立地使用 Change 或 Browse 方式显示。

如果 Browse 带 LEDIT(或 REDIT)子句, 其左(或右)部分以 Change 方式显示。两个都带上, 则两部分都以 Change 方式显示。可以通过选择 Browse 菜单基上的 Browse 和 Change 选项切换为不同的方式。

(9)LOCK <expN2>。该子句锁定在字段列表中前面的 <expN2> 个字段在 Browse 窗口上固定显示, 窗口内容左右浏览时, 它们将保持在屏幕上的位置不变, 其余字段则会随着移动。(10)LPARTITION. 一个 Browse 窗口可通过带 PARTITION 子句的 Browse 命令分裂成左右两部分。默认时, 光标将落在右部分的第一个字段编辑域上。用本子句将使光标落在左部分的第一个编辑域上。但如果使用未带 PARTITION 而带了 LPARTITION 子句的 Browse 命令, 光标将仍落在右边部分。

(11)NOAPPEND. 该子句使你不能按 Ctrl+N 或选

Browse 菜单中的 Append Record 项向数据库追加记录。注意: 可以在 VALID、WHEN 或 ON KEY LABEL 等子句带的用户自定义函数中向数据库追加记录。

(12)NOCLEAR. 使用了该子句, 当 Browse 窗口被关闭后, 其窗口图像仍保留在屏幕上显示, 但它不是一个活动的 Browse 窗口, 可用 CLEAR 命令清除。它只适用于 FoxPro for DOS。

(13)NODELETE. 默认时, 当前记录可通过按 Ctrl+T、在 Browse 菜单上选择 Toggle Delete、或用鼠标点按记录的最左列作删除标记。本子句阻止给记录作删除标记。但它不能防止你用 VALID、WHEN 或 ON KEY LABEL 等带的自定义函数给数据库记录作删除标记。

(14)NOEDIT 或 NOMODIFY. 包含 NOEDIT 或 NOMODIFY 子句可防止数据库记录被编辑修改, 仅能被浏览或搜索。但可追加或删除记录, 除非同时包含了 NOAPPEND 和 NODELETE 子句。(15)NOLGRID 和 NORGRID. 默认 Browse 窗口的字段间以垂直制表线分隔开。包含 NOLGRID 或 NORGRID 子句则取消 Browse 窗口的左或右部分的字段分隔线。取消这些垂直分隔线可避免在某些汉字系统下可能的汉字数据显示混乱的问题。

(16)NOLINK. 默认时, Browse 窗口的左右两部分是相关联的, 两部分的记录将同步滚动。用 UNLINK 子句可取消两者的联系, 这样左右两部分可各自独立操作, 比如一方移动记录指针, 另一方不会跟着移动。

(17)NOMENU. 包含 NOMENU 子句的 Browse 命令将不激活 Browse 菜单基及其选项。

(18)NOOPTIMIZE 和 NOREFRESH. NOOPTIMIZE 禁止 Rushmore 优化技术发挥作用; NOREFRESH 禁止 Browse 窗口刷新, 正常时 Browse 窗口以用 SET REFRESH 命令设置的速度刷新。NOREFRESH 常用于只读文件及特殊场合。

(19)NORMAL. 在用户自定义窗口中打开 Browse 窗口时, 它将沿用用户定义的窗口特性, 如颜色、大小、位置、标题和控制钮 (GROW、FLOAT、ZOOM 等)。如包含本子句, Browse 窗口将用默认的设置正常打开, 不受用户自定义窗口的影响。

(20)NOWAIT. 在程序中的 Browse 命令包含本子句, 该命令执行后, 不等待窗口的关闭, 马上执行后续命令。但在命令窗口单独执行 Browse 命令时, 该子句无效。

(21)PARTITION <expN3>。用本子句将 Browse 窗口分裂成两部分。数字表达式<expN3> 标明从窗口的哪一行开始分裂。

【例 3.9】从 Browse 窗口的第 30 列开始将窗口分裂成两部分,左部分以 CHANGE 方式显示。两部分各自独立,右部分中记录指针移动时,左部分记录不作同步移动;左部分的字段垂直分隔线被取消;菜单棒也不提供 Browse 菜单基;数据只能显示不能编辑;退出 Browse 窗口后,其窗口图像仍保留在屏幕上,但可用 CLEAR 命令清除。

```
CLOSE DATABASE
```

```
USE customer
```

```
Browse PARTITION 30 LEDIT NOLINK;
```

```
NOLGRID NOMENU NOEDIT NOCLEAR
```

(22)PREFERENCE <expC3>。本子句优先恢复指定名称的 Browse 窗口配置属性。不同于 LAST 子句,它可无限地恢复 FOXUSER 文件中的任意一条指定的 Browse 窗口属性记录,保存并关闭 Browse 窗口时,窗口属性同时被保存。

(23)REST。默认时, Browse 将使记录指针放到数据库的顶部;使用本子句, Browse 窗口打开时,不改变当前记录指针的位置。

(24)SAVE。该子句使 Browse 窗口及其备注字段编辑窗口保持活动和可见。在用键盘或鼠标切换到其它窗口操作后,仍可切回原 Browse 窗口。它也只在程序中发挥作用,在命令窗口使用无效。

(25)TIMEOUT <expN4>。该子句指定 Browse 窗口等待数据输入的等待时间。当 FOXPRO 等待<expN4>指定的时间后仍无输入,将自动关闭 Browse 窗口。它也只在程序中起作用。

(26)TITLE <expC4>。指定 Browse 窗口的标题。默认时,为数据库文件名,或在使用了 Browse WINDOW <winname> 且该窗口有一个标题时,沿用该窗口标题。用本子句改变之。

【例 3.10】修改在用户定义窗口中的 Browse 窗口的标题:

```
DEFINE WINDOW mybrowse FROM 1,1 TO 24,40;
```

```
CLOSE GROW COLOR SCHEME 10 TITLE '用户窗口
```

```
CLOSE DATABASES
```

```
USE customer
```

```
Browse WINDOW mybrowse;
```

```
FIELDS ytdpurch :H = '购买品';
```

```
company :H = '购买单位' TIMEOUT 3
```

```
Browse WINDOW mybrowse TITLE '销售记录编辑窗口';
```

```
FIELDS ytdpurch :H = '购买品';
```

```
company :H = '购买单位' TIMEOUT 3
```

```
RELEASE WINDOW mybrowse
```

(27)VALID [:F] <expL2> [ERROR <expC5>]。记录级校验。只有在当前记录被修改了,才执行 VALID 子句,如该记录未被修改,则忽略 VALID 子句。当然也可用 [:F]强制校验。VALID 返回真值时,可移动光标到另一条记录;返回为假,则光标不能移出该记录,并显示一错误信息 "Invalid input",可用 ERROR 子句指定自己的错误提示信息;返回为零,光标不能移出当前记录,但不显示错误信息。

【注意】前面讲的 FIELDS 子句中的[:V]选项是字段级的校验,本子句则是记录级校验。

(28)WHEN <expL3>。记录级编辑权验证。在光标移进某记录前,先执行 WHEN 子句,如果<expL3> 返回值为真,则可编辑该记录;返回值为假或 0,光标可移到该记录,但不能编辑该记录。

有关 VALID 和 WHEN 子句的设计可参阅后续讲座。

(29)WIDTH <expN5>。本子句限制在 Browse 窗口中的各字段显示的最大宽度,超出部分不显示,但可用滚动光标编辑和显示未显示的部分。

(30)WINDOW <winname1>。让 Browse 窗口继承 winname1 窗口的属性。在 FoxPro for MS-DOS 下,如果该窗口定义时有脚标题,则在 Browse 状态下,脚标题所在位置被 Browse 窗口的滚动条所取代。

(31)IN [WINDOW] <winname2> 和 IN SCREEN。指定 Browse 窗口打开到父窗口 <winname2> 或 FOXPRO 主窗口上。

(32)COLOR SCHEME <expN6> COLOR <color pair list>。指定 Browse 窗口颜色。

3. Browse 的应用

从上述介绍和实例可以看出, Browse 的功能十分丰富强大,使用简便,即使是初学者,只需灵活选择和组合功能子句,即可设计出美观实用的各种数据库编辑查询功能来。

在 FOXPRO 环境及相应的各种应用系统中, Browse 命令是使用相当普遍的数据库操作命令。(待续)